26 Chapter: Generators and Relations

Groups defined using generators and relations can be easily created in GAP. If you want to create an n-generated group, start with a free group on n generators. Then create the group by "moding out by" the relations. For example, D_4 is a 2-generated group with relations $a^4 = b^2 = (ab)^2 = e$, where a and b are the generators. [Gallian, Chapter 26, Examples 2 and 3] The following creates the group D_4 in GAP:

```
gap> f:=FreeGroup(2);
<free group on the generators [ f1, f2 ]>
gap> d4:=f/[f.1^4, f.2^2, (f.1*f.2)^2];
<fp group on the generators [ f1, f2 ]>
gap> Elements(d4);
[ <identity ...>, f2, f1^3*f2, f1, f1^3, f1*f2, f1^2*f2, f1^2 ]
```

The first command above creates a free group with two generators. The element `f.1` denotes the first generator and `f.2` denotes the second generator in the free group `f`:

```
gap> f.1;
f1
gap> f.2;
f2
```

The line `gap> d4:=f/[f.1^4, f.2^2, (f.1*f.2)^2];` creates D_4 as the free group on two generators mod the relations `f.1^4` = `f.2^2` = `( f.1 * f.2)^2`= e.

We can now use the group theory GAP commands discussed in previous chapters on `d4`. For example:

```
gap> IsAbelian(d4);
false
gap> Size(d4);
8
gap> Center(d4);
Group([ f1^2 ])
```

In addition, the `Factorization` command (see Chapter 5 of this manual) can be used to express the image of a word in the free group as an element in the factor group:

```
gap> a:=d4.1;;
gap> b:=d4.2;;
gap> Factorization(d4, a*b*a^3*b*a^5);
x1^-1
```

The first two above commands assign the letters a and b to the image of the two generators of the free group. The `Factorization` output tells us that aba^3ba^5 reduces to $a^{-1}(= a^3)$ in D_4. (GAP will denote the ith generator of the factor group by `xi`.)

You can also use GAP to help you classify a group that is defined using generators and relations. For example consider the group G defined by $G =< a, b \mid a^3 = b^9 = e, a^{-1}ba = b^{-1} >$. [Gallian, Chapter 26, Example 6]. Note that the relation $a^{-1}ba = b^{-1}$ can be rewritten as $a^2bab = e$. The below GAP commands create this group G:

```
gap> f:=FreeGroup(2);
<free group on the generators [ f1, f2 ]>
gap> G:=f/[f.1^3, f.2^9, f.1^2*f.2*f.1*f.2];
<fp group on the generators [ f1, f2 ]>
gap> Size(G);
3
```

Since there is only one group (up to isomorphism) of order 3, G must be the cyclic group of order 3.

Exercises

26.1 Use GAP to show that $< a, b \mid a^5 = b^2 = e, ba = a^2b >$ is isomorphic to $\mathbf{Z}_2$. [Gallian, Chapter 26, Exercise 4]

26.2 Let $G = < a, b \mid a^2 = b^4 = e, ab = b^3a >$.
a) Without using GAP express $a^3b^2abab^3$ in the form a^ib^j. Check your work using GAP.
b) Without using GAP express b^3abab^3a in the form a^ib^j. Check your work using GAP. [Gallian, Chapter 26, Exercise 12]

26.3 Let $G = < a, b \mid a^8 = b^2 = e, baba^3 = e >$.
a) Use GAP to find $|G|$.
b) Find the order of ab.
c) Find the center of G. [Gallian, Chapter 26, Exercise 15]

26.4 Let $G = < a, b \mid a^6 = b^3 = e, b^{-1}ab = a^3 >$. Use GAP to help you determine to which familiar group G is isomorphic. [Gallian, Chapter 26, Exercise 21]

26.5 Let $G = < a, b, c, d \mid ab = c, bc = d, cd = a, da = b >$. Use GAP to help you determine to which familiar group G is isomorphic.

26.6 Let $X_n = < a, b \mid a^n = b^2 = e, ab = ba^2 >$.
a) Find the order of X_n when $n = 3, 6, 24$ and 300.
b) Make a conjecture about the isomorphism type of X_n when n is a multiple of 3.
c) Make a conjecture about the isomorphism type of X_n when n and 3 are relatively prime. (First find the order of X_n for many appropriate values of n to help you formulate the conjecture.)

26.7 Let $G = < a, b \mid a^3 = b^3 = (ab)^2 = e >$. Use GAP to help you determine to which familiar group G is isomorphic.

Recall the command `IsomorphismGroups(G, H);` computes an isomorphism between the groups G and H. (If they are not isomorphic the command returns `fail`.) This command can also be used on groups defined using generators and relations.

```
gap> f:= FreeGroup(1);
<free group on the generators [ f1 ]>
gap> G:= f/[f.1^6];
<fp group on the generators [ f1 ]>
gap> H:= Subgroup(SymmetricGroup(6), [(1,2,3,4,5,6)]);
Group([ (1,2,3,4,5,6) ])
gap> K:= SymmetricGroup(3);
Sym( [ 1 .. 3 ] )
gap> IsomorphismGroups(H,K);
fail
gap> IsomorphismGroups(H,G);
[ (1,4)(2,5)(3,6), (1,3,5)(2,4,6) ] -> [ f1^-3, f1^2 ]
```

